Exporting data

Created using Maple 14.01

Jake Bobowski

```
> restart;
  with(StringTools) :
  FormatTime("%m−%d−%Y, %H:%M");
```

"10-03-2013, 21:18" **(1)**

This tutorial will show how to use Maple for some basic processing of digital images. First, load the "ImageTools" package.

```
> with(ImageTools) :
```

The "Read" command can be used to import a digital image, like a photo from a digital camera. Once imported, the "View" command is used to open a window that displays the image.

```
> img
      := Read("G:\\UBCO\\2013-2014\\PHYS 331\\projects\\Interferometer Optics\\30min.
    JPG");
  View(img);
```

$$img := \begin{bmatrix} 1..2592 \times 1..3872 \times 1..3 \text{ Array} \\ \text{Data Type: } float_8 \\ \text{Storage: rectangular} \\ \text{Order: C\_order} \end{bmatrix} \quad \textbf{(2)}$$

To crop unwanted sections of the image, the command "GetSubImage" can be used. It requires five arguments: 1. name of image to be cropped, 2. the number of the first row to keep, 3. the number of the first column to keep, 4. the height of the image (number of rows), and 5. the width of the image (number of columns).

```
> subImg := GetSubImage(img, 1000, 800, 1592, 2500);
  View(subImg);
```

$$subImg := \begin{bmatrix} 1000..2591 \times 800..3299 \times 1..3 \text{ Array} \\ \text{Data Type: } float_8 \\ \text{Storage: rectangular} \\ \text{Order: C\_order} \end{bmatrix} \quad \textbf{(3)}$$

Often, when analyzing an image, one may wish to convert the image colours to a gray scale. To do so, use "RGBtoGray". This command assigns a value between 0 (black) and 1 (white) to each pixel in the image.

```
> imgGray := RGBtoGray(subImg);
  View(imgGray);
```

(4)

$$imgGray := \begin{bmatrix} 1000..2591 \text{ x } 800..3299 \text{ Array} \\ \text{Data Type: } float_8 \\ \text{Storage: rectangular} \\ \text{Order: C\_order} \end{bmatrix} \quad (4)$$

The modified image can be written to a file using the "Write" command.

```
> Write("G:\\UBCO\\2013-2014\\PHYS 331\\projects\\Interferometer Optics\\30min-Gray.JPG",
      imgGray) :
```

Often, Now, let's see how to do some simple analysis of the modified image. First, individual pixels can be accessed using "img[row number, column number]".

```
> imgGray[1000, 800];
  imgGray[1750, 2500];
```

$$0.00392156862745098$$
$$0.210078431372549 \quad (5)$$

The total number of rows and columns in the modfied image can be extracted using "Height(img)" and "Width(img)" respectively.

```
> Height(imgGray);
  Width(imgGray);
```

$$1592$$
$$2500 \quad (6)$$

We can use loops to sum the total `brightness' of the image pixels. The average pixel `brightness' is obtained by dividing the sum by the product of the width and the height.

```
> intensity := 0 :
  for i from 800 to 800 + Width(imgGray) − 1 do:
    for j from 1000 to 1000 + Height(imgGray) − 1 do:
      intensity := intensity + imgGray[j, i];
    end do:
  end do:
  intensity;
  meanValue := intensity / (Width(imgGray)·Height(imgGray));
```

$$1.80246874037590\ 10^5$$
$$meanValue := 0.0452881593059271 \quad (7)$$

Maple has an "Intensity" function that can be used to some analysis without requiring use to write for loops. For example, here is the mean value that we just calculated.

```
> Intensity(imgGray, mean);
```

$$0.04528815930 \quad (8)$$

Here are some additional examples of using "Intensity" for quantative analysis of images.

```
> Intensity(imgGray, rms);
  Intensity(imgGray, variance);
```

```
Intensity(imgGray, range);
Intensity(imgGray, maximum);
```

$$0.08684379158$$

$$0.005490826765$$

$$0. ..0.929882352941176382$$

$$0.929882352941176382 \quad \textbf{(9)}$$

```
>
```